Using computer algebra to check if a family of algebras are pair-wise non-isomorphic Given an infinite field $k$, consider a quiver $\Gamma$ with one vertex and two arrows $x,y$ and define $R=k\Gamma/(x,y)^2.$ This is a three-dimensional $k$-algebra.
Now consider the additive group of 2 by 2 matrices over $R$ denoted by $M_2(R)$. To define the multiplication, consider matrices $$X_t=\begin{bmatrix}
x & ty \\
y & x
\end{bmatrix}$$ with $t\in k$. Then we can introduce a multiplication $A\cdot B= AX_tB,$ where on the right we have the classic matrix product. We will call $M_2(R)$ with this multiplication $H_t$. For any $t\in k$ the $H_t$ is a non-unital $k$-algebra. I am interested in whether these algebras are pairwise non-isomorphic.
I am aware of the 1-parameter family of non-isomorphic algebras $k\{x,y\}/(x,y,xy-tyx)$. For this family, it is possible to brute-force the proof of the fact that they are actually non-isomorphic. By the way, if any of you are aware of some other ways of proving this fact, please let me know!
Now the dimension of $H_t$ is 12, so looking at all linear maps between these algebras and checking whether they are homomorphisms by hand seems impossible. I also couldn't come up with any high-brow arguments for why these algebras must be non-isomorphic. I did, however, come up with the following:
Denote the matrix with a $1$ in the $(i,j)$ position and 0 everywhere else by $E_{ij}$, and say we have a linear map $F: H_t\to H_m$.If we want $F$ to be a homomorphism, we must have $$F(E_{11})F(E_{21})=tF(E_{12})F(E_{11}),\ F(E_{11})F(E_{21})=tF(E_{12})F(E_{11}).$$
Notice that any matrix with only linear combinations of $x,\ y$ as elements is annihilated by any other element of $H_t$, so without loss of generality, we may assume that $F(E_{ij})$ is a linear combination of $E_{ij}$. So if it is a homomorphism, $F$ is completely defined by the images of $E_{ij}$, which are each defined by 4 scalars.
When we write $F(E_{ij})$ as linear combinations of $E_{ij}$, the above system of 2 equations turns into 16 homogeneous equations of order 2 in terms of 12 scalar variables with $t$ and $m$ as parameters. If we could show this system is inconsistent (for general $t$ and $m$), we would have that there are actually no non-zero homomorphisms $H_t \to H_m$. This is also the source of my hypothesis that these algebras are non-isomorphic: each homomorphism is defined by 16 scalars subject to a seemingly very large system of homogeneous equations of second order.
To show that this system is inconsistent I decided to pick a subsystem of 12 equations in 12 variables and compute the Macaulay resultant (or check if it's nonzero, which would mean that it's a polynomial in $t,m$ and we would have no homomorphisms for "generic" $t,m$). I tried doing it in Macaulay2 using this code:
loadPackage "EliminationMatrices"
R= QQ[t,m,a,b,c,d,u,v,w,r,p,q,z,s]
l={a,b,c,d,u,v,w,r,p,q,z,s}
f1=u*p+v*z-t*(a*u+b*w)
f2=u*s+v*q-t*(a*v+b*r)
f3=w*p+r*z-t*(c*u+d*w)
f4=w*s+r*q-t*(c*v+d*r)
f5=v*p+m*u*z-t*(b*u+m*a*w)
f6=v*s+m*u*q-t*(b*v-m*a*r)
f7=r*p+m*w*z-t*(d*u+m*c*w)
f8=t*s+m*w*q-t*(d*v+m*c*r)
f9=u*u+v*w-a*p-b*z
f10=u*v+v*r-a*s-b*q
f11=w*u+r*w-c*p-d*z
f12=w*v+r^2-c*s-d*q
M=matrix{{f1,f2,f3,f4,f5,f6,f7,f8,f9,f10,f11,f12}}
MR=elimanationMatrix(l,M)

This should give me a matrix of full row-rank if and only if the resultant is zero. Here the variables in $l$ are just the coordinates of the $F(E_{ij})$ in the basis of $E_{ij}$. Now I know that the complexity of this code is exponential in the number of variables and am starting to doubt that this code will ever finish executing.
I am also very unfamiliar with Macaulay2, this is only my second time using it. This code has been running for more than 24 hours now. So now I finally come to the question: is the above code inefficient? Or are there alternative ways to check if the resultant is zero? Or hopefully even a non-computational way of solving this problem?
 A: For this and other Macaulay2-related questions, I highly recommend the Macaulay2 google group.
In general, there are some "exploratory" techniques (i.e. not quite a proof) which are more efficient and may indicate whether your original approach will yield the desired outcome. For your example, I think it makes sense for starters to work with coefficients over a large prime field and plug in random constants for $m$ and $t.$
In the example below, I take $I$ to be the ideal $\langle f_1, \ldots , f_{12}, f_{13} \rangle $ in the polynomial ring $\mathbb{F}_{10007}[a,b,c,d,u,v,w,r,p,q,z,s],$ where $f_{13}$ is a random inhomogeneous polynomial of degree $1$ (this can be interpreted as random chart on the projective space $\mathbb{P}^{11}.$)
The computation that $\dim I = 3$ takes about 1s on my laptop.
This suggests (but doesn't quite prove) that $V_{\mathbb{C} \, \mathbb{P}^{11}} (f_1, \ldots , f_{12}) \ne \emptyset .$
Your conjecture may still be correct---as you mention, this is only a subsystem of the one you've derived...
pr = 10007
FF = ZZ/pr
t = random FF
m = random FF
R = FF[a,b,c,d,u,v,w,r,p,q,z,s]
f1 = u*p+v*z-t*(a*u+b*w)
f2 = u*s+v*q-t*(a*v+b*r)
f3 = w*p+r*z-t*(c*u+d*w)
f4 = w*s+r*q-t*(c*v+d*r)
f5 = v*p+m*u*z-t*(b*u+m*a*w)
f6 = v*s+m*u*q-t*(b*v-m*a*r)
f7 = r*p+m*w*z-t*(d*u+m*c*w)
f8 = t*s+m*w*q-t*(d*v+m*c*r)
f9 = u*u+v*w-a*p-b*z
f10 = u*v+v*r-a*s-b*q
f11 = w*u+r*w-c*p-d*z
f12 = w*v+r^2-c*s-d*q
f13 = random(1,R)-1
G = groebnerBasis(ideal(f1,f2,f3,f4,f5,f6,f7,f8,f9,f10,f11,f12,f13), Strategy=>"F4");
dim ideal leadTerm G

A: There's a practical computer-algebra way to determine whether
(a) they're all isomorphic up to finitely many exceptions, or
(b) there "(bounded finite)-to-one" non-isomorphic, i.e., there exists $n$ such that for each $t$ the set of $s$ such that $A_s\simeq A_t$ has cardinal $\le n$.
See this question.
By practical, I mean it remains doable in reasonable time for algebras of higher dimension, say $\le 100$.
